

Physics

HP COMPUTER CURRICULUM

Geometrical Optics

TEACHERS ADVISOR

HEWLETT  PACKARD

Hewlett-Packard
Computer Curriculum Series

physics
TEACHER'S ADVISOR

geometrical
optics

by Herbert D. Peckham
Gavilan College

edited by Briana Burns and Jean Danver
Hewlett-Packard

Hewlett-Packard Company
11000 Wolfe Road
Cupertino, California 95014

© Copyright, 1972, by
HEWLETT-PACKARD COMPANY
Cupertino, California
Printed in the U.S.A.

The Hewlett-Packard Computer Curriculum Series represents the published results of a Curriculum Development project sponsored by the Data Systems Division of Hewlett-Packard Company. This project is under the directorship of Jean H. Danver.

This material is designed to be used with any Hewlett-Packard system with the BASIC programming language such as the 9830A, Educational BASIC, and the 2000 and 3000 series systems.

The Hewlett-Packard Company makes no warranty expressed or implied, and assumes no responsibility in connection with the operation of the program material contained herein.

Copyright © 1972 by Hewlett-Packard Company, Cupertino, California. All rights reserved. No part of this publication may be reproduced, stored in a retrieval system (e.g., in memory, disc or core) or be transmitted by any means, electronic, mechanical, photocopy, recording or otherwise, without prior written permission from the publisher.

Printed in the U.S.A.

TABLE OF CONTENTS

5	INTRODUCTION
9	REVIEW OF COORDINATE GEOMETRY
9	Exercise 1 – Coordinates
10	Exercise 2 – Distance Between Points
11	FERMAT'S PRINCIPLE
11	Reflection
11	Exercise 3 – Rays Between Fixed Points
12	Exercise 4 – Rays Between Changing Points
12	Exercise 5 – Precision Computation
14	Exercise 6 – Angles of Incidence & Reflection
15	Exercise 7 – Discovery
16	Exercise 8 – Generalization
16	Exercise 9 – Corner Reflector
18	Refraction
18	Exercise 10 – Refraction With Precision Computation
19	Exercise 11 – Refraction Between Different Media
21	Exercise 12 – Angles of Incidence and Refraction
23	Exercise 13 – Discovery
25	Exercise 14 – Generalization
25	Exercise 15 – Double Refraction
26	Exercise 16 – A Complex Problem
29	RAY TRACING
29	Exercise 17 – A Single Lens
29	Exercise 18 – Two Lenses
30	Exercise 19 – A Program For N Lenses
31	Exercise 20 – Intersection of Rays
31	Exercise 21 – Description of Image
33	OPTIONAL MATERIAL
33	Exercise 22 – A Matrix Program
34	Exercise 23 – Description of Image

NOTES

INTRODUCTION

This Physics Set of the Hewlett-Packard Computer Curriculum Series consists of a set of a Student Lab Book and a corresponding Teacher's Advisor. It was designed to help meet the need for computer-oriented problems in physics, providing students an opportunity to use a computer as a problem solving tool within a particular subject matter area.

The materials are designed for flexible use as desired by the individual instructor. The material and exercises in this unit are intended as an "enrichment" experience in the field of geometrical optics. Fermat's principle is used to discover the laws of reflection and refraction. Normally this requires differential calculus. However, with the speed of the computer, the same task requires only algebra and some trigonometry and therefore can be used at the introductory college level and in high school. The final topic covered is the method of ray tracing as applied to thin lenses. Since neither of the topics above is usually covered in the introductory texts, the use of this material will not compete with your text. Instead, it can be used to supplement and enrich in any fashion you choose.

The degree of difficulty of the material is dependent upon the amount of assistance which you choose to provide. With no assistance, the better physics student should be challenged. However, given a good deal of assistance, any physics student should be able to work out the exercises with no great difficulty. The level of the material is determined by the assumption that students taking introductory physics will be quite capable as a group.

The Lab Book provides text material and programming exercises for the students, a sample program and advanced problems. The Teacher's Advisor contains an example of a program to solve each exercise and a brief discussion of the important elements of the exercise.

For best results, you should study all the solutions until you are certain you have a complete grasp of the general methods. This should be done before assigning the material to the class. Generally, the exercises are cumulative so that as techniques are developed they are used in subsequent exercises. Therefore, you will probably wish to proceed through the exercises in the order in which they are given.

You will undoubtedly think of different programming methods or techniques as you study the example programs. Encourage the students to do the same. There are no *approved* solutions. All solutions are acceptable if they produce the correct results. At this level, there is no need for emphasis on the efficiency of a student's program. The important question is, does it work?

NOTES

REVIEW OF COORDINATE GEOMETRY**Exercise 1 – Coordinates**

This is designed solely to provide practice in the location of points on a two-dimensional cartesian coordinate system. If your students do not need this practice, go on to Exercise 2. The solution follows.

Point	x	y
A	1	1
B	0	-2
C	-4	0
D	0	0
E	5.5	0.5
F	3	4.5
G	-2	3
H	-4.5	-3
I	2.5	-4
J	0	5

Exercise 2 – Distance Between Points

The only question to be decided in this exercise is how to control the looping through the data. The program below uses a FOR loop. Other methods are possible. In line 110, N is the number of line segments to be processed. Input of each segment takes place in line 130 with the computation in line 140.

```
100  REM DISTANCE BETWEEN TWO POINTS
110  READ N
120  FOR I=1 TO N
130  READ X1,Y1,X2,Y2
140  LET D=SQR((X2-X1)^2+(Y2-Y1)^2)
150  PRINT D
160  NEXT I
800  DATA 10
801  DATA 1,1,5,6
802  DATA -5,2,1,-2
803  DATA -3.5,-1.2,2.8,4.1
804  DATA 0,0,0,1.4283
805  DATA 0,-6.732,0,0
806  DATA -2,-3,-3,-2
807  DATA .923,-.149,-.758,-.358
808  DATA 0,1.414,1.414,0
809  DATA 2.8,1.5,-4.1,-2.1
810  DATA 0,3.6,0,-2.4
999  END
```

RUN

```
6.40312
7.2111
8.23286
1.4283
6.732
1.41421
1.69394
1.9997
7.78267
6
```

FERMAT'S PRINCIPLE

REFLECTION

Exercise 3 – Rays Between Fixed Points

The index of refraction is set equal to 1 in line 90. This is not required in this particular problem, but will be needed in subsequent ones. Hence, we establish the pattern here. In lines 100 through 130 the optical path P_0 is computed for the first integer value of X , the value of $X = 1$ being assigned at the same time to the variable X_0 which will be developed as the reflection point. A new optical path is computed in lines 150 through 170 each time through the loop. In lines 180 through 190 the new optical path is compared to the smallest of the previous optical paths. If the new one is smaller, we update P_0 and X_0 .

```

90  LET N=1
100 LET D1=SQR((1-1)^2+(0-4)^2)
110 LET D2=SQR((11-1)^2+(6-0)^2)
120 LET P0=N*D1+N*D2
130 LET X0=1
140 FOR X=2 TO 11
150 LET D1=SQR((X-1)^2+(0-4)^2)
160 LET D2=SQR((11-X)^2+(6-0)^2)
170 LET P1=N*D1+N*D2
180 IF P1>P0 THEN 210
190 LET P0=P1
200 LET X0=X
210 NEXT X
220 PRINT "REFLECTION AT X = "X0
999 END

```

RUN

REFLECTION AT X = 5

Exercise 4 – Rays Between Changing Points

This is a simple modification of the program in Exercise 3. The initial and final points are read into the program in line 120. The appropriate changes are then carried through the balance of the program.

```
100 LET N=1
120 INPUT X1,Y1,X2,Y2
130 LET D1=Y1
140 LET D2=SQR((X2-X1)^2+Y2^2)
150 LET P0=N*D1+N*D2
160 LET X0=X1
170 FOR X=X1+1 TO X2
180 LET D1=SQR((X-X1)^2+Y1^2)
190 LET D2=SQR((X2-X)^2+Y2^2)
200 LET P1=N*D1+N*D2
210 IF P1>P0 THEN 240
220 LET P0=P1
230 LET X0=X
240 NEXT X
250 PRINT "REFLECTION AT X = "X0
999 END
```

RUN

```
?0,20,10,20
REFLECTION AT X = 5
```

Exercise 5 – Precision Computation

S (in line 100) is the initial step size in the search. In line 110, M (number of significant digits in answer desired) is set. The search procedure in lines 120 through 280 is the same as in Exercise 4. The only difference is found in line 180 which sets a counter C used to test against M. The value of the counter is tested in line 290. If less than M the search limits are reset in lines 300 through 330. Then the program loops back to the search procedure.

```
100 LET S=1
110 INPUT M
120 LET N=1
130 INPUT X1,Y1,X2,Y2
140 LET D1=Y1
150 LET D2=SQR((X2-X1)^2+Y2^2)
160 LET P0=N*D1+N*D2
170 LET X0=X1
180 LET C=1
190 LET A=X1+1
200 LET B=X2
210 FOR X=A TO B STEP S
220 LET D1=SQR((X-X1)^2+Y1^2)
230 LET D2=SQR((X2-X)^2+Y2^2)
240 LET P1=N*D1+N*D2
250 IF P1>P0 THEN 280
260 LET P0=P1
270 LET X0=X
280 NEXT X
290 IF C >= M THEN 350
300 LET A=X0-S
310 LET B=X0+S
320 LET S=S/10
330 LET C=C+1
340 GOTO 210
350 PRINT "REFLECTION AT X = "X0
999 END
```

RUN

```
?4
?0,10,10,8
REFLECTION AT X = 5.56101
```

RUN

```
?4
?-10,10,10,10
REFLECTION AT X = -1.81608E-08
```

RUN

```
?4
?5,10,10,15
REFLECTION AT X = 7.00401
```

Exercise 6 – Angles of Incidence and Reflection

The procedure in lines 100 through 340 is the same as in Exercise 5. Angles of incidence and reflection are computed in lines 350 and 360. These angles are converted to degrees and output in lines 370 and 380.

```
100 LET S=1
110 INPUT M
120 LET N=1
130 INPUT X1,Y1,X2,Y2
140 LET D1=Y1
150 LET D2=SQR((X2-X1)^2+Y2^2)
160 LET P0=N*D1+N*D2
170 LET X0=X1
180 LET C=1
190 LET A=X1+1
200 LET B=X2
210 FOR X=A TO B STEP S
220 LET D1=SQR((X-X1)^2+Y1^2)
230 LET D2=SQR((X2-X)^2+Y2^2)
240 LET P1=N*D1+N*D2
250 IF P1>P0 THEN 280
260 LET P0=P1
270 LET X0=X
280 NEXT X
290 IF C >= M THEN 350
300 LET A=X0-S
310 LET B=X0+S
320 LET S=S/10
330 LET C=C+1
340 GOTO 210
350 LET I=ATN((X0-X1)/Y1)
360 LET R=ATN((X2-X0)/Y2)
370 PRINT "ANGLE OF INCIDENCE = "57.2957*I"DEGREES"
380 PRINT "ANGLE OF REFLECTION ="57.2957*R"DEGREES"
999 END
```

RUN

```
?4
?0,10,20,5
ANGLE OF INCIDENCE = 53.1335 DEGREES
ANGLE OF REFLECTION = 53.1232 DEGREES
```

Exercise 7 – Discovery

The program in Exercise 6 is modified here to process several problems. The number of problems Z is input in line 95. The problem loop is established in lines 96 and 400. Inside this loop, the details are the same as in Exercise 6.

```
95  READ Z
96  FOR J=1 TO Z
100  LET S=1
110  LET M=3
120  LET N=1
130  READ X1,Y1,X2,Y2
140  LET D1=Y1
150  LET D2=SQR((X2-X1)2+Y22)
160  LET P0=N*D1+N*D2
170  LET X0=X1
180  LET C=1
190  LET A=X1+1
200  LET B=X2
210  FOR X=A TO B STEP S
220  LET D1=SQR((X1-X)2+Y12)
230  LET D2=SQR((X2-X)2+Y22)
240  LET P1=N*D1+N*D2
250  IF P1>P0 THEN 280
260  LET P0=P1
270  LET X0=X
280  NEXT X
290  IF C >= M THEN 350
300  LET A=X0-S
310  LET B=X0+S
320  LET S=S/10
330  LET C=C+1
340  GOTO 210
350  LET I=ATN((X0-X1)/Y1)
360  LET R=ATN((X2-X0)/Y2)
370  LET I=57.2957*I
380  LET R=57.2957*R
390  PRINT X1;Y1;X2;Y2;I,R
400  NEXT J
800  DATA 5
801  DATA -10,5,8,4
802  DATA 2,10,15,10
803  DATA 0,5,10,10
804  DATA 20,25,30,15
805  DATA -20,6,-10,10
999  END
```

RUN

-10	5	8	4	63.4577	63.4062
2	10	15	10	33.0238	33.0238
0	5	10	10	33.6636	33.7032
20	25	30	15	14.0362	14.0363
-20	6	-10	10	32.0051	32.0055

Exercise 8 – Generalization

In this exercise, the student is asked to “discover” the law of specular reflection. Note carefully that nothing in our development mentions the fact that the incident and reflected rays lie in the same plane. This certainly has been implied since we have been working in a coordinate plane. As an optional exercise, students might be asked to repeat the development above in a three-dimensional coordinate system in which possible paths outside of a single plane could be investigated. Of course the student will discover that the rays do lie in a single plane.

Exercise 9 – Corner Reflection

This should be considered an optional exercise for the better student. Since two reflections are involved, a double loop is required to conduct the search. Trace the program carefully to see how the search is carried out. The object is to let the student discover the characteristics of a corner reflector. Have the students plot their results and discover that after two reflections, the ray is returning parallel to the initial ray.

```

100 LET S=1
110 INPUT M
120 LET N=1
130 INPUT X1,Y1,X2,Y2
140 LET D1=X1
150 LET D2=Y1
160 LET D3=SQR(X2^2+Y2^2)
170 LET P0=N*(D1+D2+D3)
180 LET X0=0
190 LET Y0=Y1
200 LET C=1

```

```

210 LET A1=Y1-1
220 LET B1=0
230 LET A2=0
240 LET B2=X2
250 FOR Y=A1 TO B1 STEP -S
260 FOR X=A2 TO B2 STEP S
270 LET D1=SQR((Y1-Y)2+X12)
280 LET D2=SQR(Y2+X2)
290 LET D3=SQR(Y22+(X2-X)2)
300 LET P1=N*(D1+D2+D3)
310 IF P1>P0 THEN 350
320 LET P0=P1
330 LET X0=X
340 LET Y0=Y
350 NEXT X
360 NEXT Y
370 IF C >= M THEN 450
380 LET A1=Y0+S
390 LET B1=Y0-S
400 LET A2=X0-S
410 LET B2=Y0+S
420 LET S=S/10
430 LET C=C+1
440 GOTO 250
450 PRINT "REFLECTION AT"
460 PRINT 0,Y0;"AND"X0,0
999 END

```

RUN

```

?3
?5,10,10,5
REFLECTION AT
0          5.          AND 5.          0

```

RUN

```

?3
?2,5,4,3
REFLECTION AT
0          2.33        AND 1.75        0

```

REFRACTION

Exercise 10 – Refraction With Precision Computation

The modifications from the program in Exercise 5 follow from the discussion in the student's Lab Book.

```
100 LET S=1
110 INPUT M
120 INPUT N1,N2
130 INPUT X1,Y1,X2,Y2
140 LET D1=Y1
150 LET D2=SQR((X2-X1)^2+Y2^2)
160 LET P0=N1*D1+N2*D2
170 LET X0=X1
180 LET C=1
190 LET A=X1+1
200 LET B=X2
210 FOR X=A TO B STEP S
220 LET D1=SQR((X-X1)^2+Y1^2)
230 LET D2=SQR((X2-X)^2+Y2^2)
240 LET P1=N1*D1+N2*D2
250 IF P1>P0 THEN 280
260 LET P0=P1
270 LET X0=X
280 NEXT X
290 IF C >= M THEN 350
300 LET A=X0-S
310 LET B=X0+S
320 LET S=S/10
330 LET C=C+1
340 GOTO 210
350 PRINT "REFRACTION AT X = "X0
999 END
```

RUN

?4
?1,1.33
?0,10,20,-10
REFRACTION AT X = 12.693

RUN

?4
?1.33,1
?0,10,20,-10
REFRACTION AT X = 7.32201

RUN

?4
?1,1
?0,10,20,-10
REFRACTION AT X = 10

Exercise 11 – Refraction Between Different Media

Following are runs of the program shown for Exercise 10 using the values given in Exercise 11.

RUN

?4
?1.48,1.33
?-10,10,10,-10
REFRACTION AT X = -1.048

RUN

?4
?1.33,1.65
?-10,10,10,-10
REFRACTION AT X = 2.085

PHYSICS

Hewlett-Packard Computer Curriculum

RUN

?4

?1.00,1.65

?-10,10,10,-10

REFRACTION AT X = 4.28901

RUN

?4

?1.65,1.00

?-10,10,10,-10

REFRACTION AT X = -4.27399

RUN

?4

?1,1

?-10,10,10,-10

REFRACTION AT X = -1.81608E-08

Exercise 12 – Angles of Incidence and Refraction

The changes in the program used for Exercise 10 are specified in the student's Lab Book.

```
100 LET S=1
110 INPUT M
120 INPUT N1,N2
130 INPUT X1,Y1,X2,Y2
140 LET D1=Y1
150 LET D2=SQR((X2-X1)^2+Y2^2)
160 LET P0=N1*D1+N2*D2
170 LET X0=X1
180 LET C=1
190 LET A=X1+1
200 LET B=X2
210 FOR X=A TO B STEP S
220 LET D1=SQR((X-X1)^2+Y1^2)
230 LET D2=SQR((X2-X)^2+Y2^2)
240 LET P1=N1*D1+N2*D2
250 IF P1>P0 THEN 280
260 LET P0=P1
270 LET X0=X
280 NEXT X
290 IF C >= M THEN 350
300 LET A=X0-S
310 LET B=X0+S
320 LET S=S/10
330 LET C=C+1
340 GOTO 210
350 PRINT "REFRACTION AT X = "X0
360 LET I=57.2957*ATN((X0-X1)/Y1)
370 LET R=57.2957*ATN((X2-X0)/Y2)
380 PRINT "ANGLE OF INCIDENCE = "ABS(I)"DEGREES"
390 PRINT "ANGLE OF REFRACTION = "ABS(R)"DEGREES"
999 END
```

PHYSICS

Hewlett-Packard Computer Curriculum

RUN

?4

?1.48,1.33

?-10,10,10,-10

REFRACTION AT X = -1.048

ANGLE OF INCIDENCE = 41.8348 DEGREES

ANGLE OF REFRACTION = 47.8504 DEGREES

RUN

?4

?1.33,1.65

?-10,10,10,-10

REFRACTION AT X = 2.085

ANGLE OF INCIDENCE = 50.3931 DEGREES

ANGLE OF REFRACTION = 38.3616 DEGREES

RUN

?4

?1.00,1.65

?-10,10,10,-10

REFRACTION AT X = 4.28901

ANGLE OF INCIDENCE = 55.0141 DEGREES

ANGLE OF REFRACTION = 29.7306 DEGREES

RUN

?4

?1.65,1.00

?-10,10,10,-10

REFRACTION AT X = -4.27399

ANGLE OF INCIDENCE = 29.7955 DEGREES

ANGLE OF REFRACTION = 54.9858 DEGREES

RUN

?4

?1,1

?-10,10,10,-10

REFRACTION AT X = -1.81608E-08

ANGLE OF INCIDENCE = 44.9999 DEGREES

ANGLE OF REFRACTION = 44.9999 DEGREES

Exercise 13 – Discovery

The program below is a simple modification of the program used for Exercise 12.
The modifications are specified in the Lab Book.

```
100 LET S=1
110 INPUT M
120 INPUT N1,N2
130 INPUT X1,Y1,X2,Y2
140 LET D1=Y1
150 LET D2=SQR((X2-X1)^2+Y2^2)
160 LET P0=N1*D1+N2*D2
170 LET X0=X1
180 LET C=1
190 LET A=X1+1
200 LET B=X2
210 FOR X=A TO B STEP S
220 LET D1=SQR((X-X1)^2+Y1^2)
230 LET D2=SQR((X2-X)^2+Y2^2)
240 LET P1=N1*D1+N2*D2
250 IF P1>P0 THEN 280
260 LET P0=P1
270 LET X0=X
280 NEXT X
290 IF C >= M THEN 350
300 LET A=X0-S
310 LET B=X0+S
320 LET S=S/10
330 LET C=C+1
340 GOTO 210
350 PRINT "REFRACTION AT X = "X0
360 LET I=ABS(ATN((X0-X1)/Y1))
370 LET R=ABS(ATN((X2-X0)/Y2))
380 PRINT "ANGLE OF INCIDENCE = "57.2957*I"DEGREES"
390 PRINT "ANGLE OF REFRACTION= "57.2957*R"DEGREES"
400 PRINT N1*I,N1*TAN(I),N1*COS(I),N1*SIN(I)
410 PRINT N2*R,N2*TAN(R),N2*COS(R),N2*SIN(R)
999 END
```

PHYSICS

Hewlett-Packard Computer Curriculum

RUN

?3

?1.48,1.33

?-10,10,10,-10

REFRACTION AT X = -1.05

ANGLE OF INCIDENCE = 41.8285 DEGREES

ANGLE OF REFRACTION= 47.8555 DEGREES

1.08047	1.3246	1.10281	.987018
---------	--------	---------	---------

1.11087	1.46965	.892431	.986137
---------	---------	---------	---------

RUN

?3

?1.33,1.65

?-10,10,10,-10

REFRACTION AT X = 2.08

ANGLE OF INCIDENCE = 50.3815 DEGREES

ANGLE OF REFRACTION= 38.3792 DEGREES

1.1695	1.60664	.848104	1.02451
--------	---------	---------	---------

1.10524	1.3068	1.29347	1.02443
---------	--------	---------	---------

RUN

?3

?1.00,1.65

?-10,10,10,-10

REFRACTION AT X = 4.28

ANGLE OF INCIDENCE = 54.9971 DEGREES

ANGLE OF REFRACTION= 29.7695 DEGREES

.959882	1.428	.573616	.819124
---------	-------	---------	---------

.857302	.9438	1.43225	.819246
---------	-------	---------	---------

RUN

?3

?1.65,1.00

?-10,10,10,-10

REFRACTION AT X = -4.27999

ANGLE OF INCIDENCE = 29.7695 DEGREES

ANGLE OF REFRACTION= 54.9971 DEGREES

.857302	.943801	1.43225	.819247
---------	---------	---------	---------

.959882	1.428	.573617	.819124
---------	-------	---------	---------

Exercise 14 – Generalization

The student should come up with some version of Snell's Law at this point.

Exercise 15 – Double Refraction

This program contains essentially the same search program as used in Exercise 10. The significant difference is that a double loop is required.

```
100 LET S=1
110 INPUT M
120 INPUT N1,N2,N3
130 INPUT X1,Y1,X2,Y2
140 INPUT C1,C2
150 LET D1=Y1-C1
160 LET D2=C1-C2
170 LET D3=SQR((X2-X1)2+(Y2-C2)2)
180 LET P0=N1*D1+N2*D2+N3*D3
190 LET X8=X1
200 LET X9=X1
210 LET C=1
220 LET A1=X1+1
230 LET B1=X2
240 LET A2=X1
250 LET B2=X2
260 FOR U=A1 TO B1 STEP S
270 FOR V=A2 TO B2 STEP S
280 LET D1=SQR((U-X1)2+(Y1-C1)2)
290 LET D2=SQR((V-U)2+(C1-C2)2)
300 LET D3=SQR((X2-V)2+(Y1-C1)2)
310 LET P1=N1*D1+N2*D2+N3*D3
320 IF P1>P0 THEN 360
330 LET P0=P1
340 LET X8=U
350 LET X9=V
360 NEXT V
370 NEXT U
380 IF C >= M THEN 460
390 LET A1=X8-S
400 LET B1=X8+S
410 LET A2=X9-S
420 LET B2=X9+S
430 LET S=S/10
440 LET C=C+1
450 GOTO 260
460 PRINT "RAY PASSES THROUGH"
470 PRINT X8;C1"AND" "X9;C2
999 END
```

```

RUN
?3
?1,1.33,1
?-10,10,10,-10
?2,-2
RAY PASSES THROUGH
-1.32      2      AND      1.33      -2

```

Exercise 16 – A Complex Problem

This program, like the one in Exercise 15, requires a double loop in the search. Essentially it is the same program, with a few geometrical differences.

```

100 LET S=1
110 INPUT M
120 INPUT N1,N2
130 INPUT X1,Y1,X2,Y2
140 LET D1=Y1
150 LET D2=-X1
160 LET D3=SQR(X2^2+Y2^2)
170 LET P0=N1*(D1+D2)+N2*D3
180 LET X0=X1
190 LET Y0=0
200 LET C=1
210 LET A1=X1+1
220 LET B1=0
230 LET A2=0
240 LET B2=Y2
250 FOR X=A1 TO B1 STEP S
260 FOR Y=A2 TO B2 STEP S
270 LET D1=SQR((X-X1)^2+Y1^2)
280 LET D2=SQR(X^2+Y^2)
290 LET D3=SQR(X2^2+(Y2-Y)^2)
300 LET P1=N1*(D1+D2)+N2*D3
310 IF P1>P0 THEN 350
320 LET P0=P1
330 LET X0=X
340 LET Y0=Y
350 NEXT Y
360 NEXT X
370 IF C >= M THEN 450
380 LET A1=X0-S
390 LET B1=X0+S
400 LET A2=Y0-S

```

```
410 LET B2=Y0+S
420 LET S=S/10
430 LET C=C+1
440 GOTO 250
450 PRINT "RAY PASSES THROUGH "
460 PRINT X0,0"AND      "0;Y0
999 END
```

```
RUN
?3
?1,1.5
?-10,10,10,10
RAY PASSES THROUGH
-2.66      0      AND      0      3.63
```

NOTES

RAY TRACING

Exercise 17 – A Single Lens

For the space:

$$\theta_2 = \theta_1 = -0.1$$

$$y_2 = y_1 + s\theta_1 = 1.3(20)(-0.1) = -3$$

This ray is then carried through the lens:

$$\theta_2 = \theta_1 - y_1/f = (-0.1) - (-3)/10 = +0.2$$

$$y_2 = y_1 = -3$$

The output ray has $\theta = 0.2$ and $y = -3$.

Exercise 18 – Two Lenses

The results of Exercise 17 are input into the space equations:

$$\theta_2 = \theta_1 = 0.2$$

$$y_2 = y_1 + s\theta_1 = (-3) + (10)(0.2) = -1$$

This ray is input into the lens equations to get the final result.

$$\theta_2 = \theta_1 - y_1/f = (0.2) - (-1)/(-20) = -0.15$$

$$y_2 = y_1 = -1$$

The output ray has $\theta = 0.15$ and $y = -1$.

Exercise 19 – A Program for N Lenses

Line 100 contains the description of the initial ray which enters the optical system. T stands for “Theta”. N1 is the number of optical elements (either a space or a lens) in the system and is used to control the limit on the FOR loop opened in line 120. Inside the loop, the data for each element is read in, tested to see which equation is appropriate, and the ray then updated. After all the elements of the system have been processed the description of the output ray is typed out.

```
100  READ Y,T
110  READ N1
120  FOR I=1 TO N1
130  READ N2,A
140  IF N2=2 THEN 170
150  LET Y=Y+A*T
160  GOTO 180
170  LET T=T-Y/A
180  NEXT I
200  PRINT Y,T
210  DATA 0,.1
220  DATA 4
230  DATA 1,10
240  DATA 2,20
250  DATA 1,20
260  DATA 2,10
999  END
```

RUN

2 -.15

Exercise 20 – Intersection of Rays

Using the program from Exercise 19 the following results are obtained.

```
210 DATA 1,0
RUN
0          -.05
```

```
210 DATA 1,-.1
RUN
-2         .1
```

Examining the results we see that the two rays differ in position by 2 units, have a relative angle between each other of 0.15, and are converging as they move away from the last element in the optical system. Thus the rays will intersect $2/0.15 = 13.33$ units to the right of the last element in the system.

Exercise 21 – Description of Image

The program from Exercise 19 is modified with new data and run with the results below:

```
220 DATA 2
230 DATA 1,20
240 DATA 2,10

210 DATA 1,0
RUN
1          -.1
```

```
210 DATA 1,-.1
RUN
-1         0
```

The two output rays are two units apart, have a relative angle between them of 0.1 and are converging. Therefore the image is *real*. The rays intersect $2/.1 = 20$ units to the right 1 unit below the optical axis. Therefore the image is *inverted*, and the same size as the object.

NOTES

OPTIONAL MATERIAL

Exercise 22 – A Matrix Program

The matrix X is used to hold the input ray. Each of the matrices corresponding to an optical element is read into the system and multiplied into M, the system matrix. Finally the output ray Y is the product of the system matrix M and the input ray X.

```
100 DIM A[2,2],M[2,2],B[2,2],X[2,1],Y[2,1]
110 MAT READ X
120 READ N
130 MAT M=IDN
140 FOR I=1 TO N
150 MAT READ A
160 MAT B=M*A
170 MAT M=B
180 NEXT I
190 MAT Y=M*X
200 MAT PRINT Y
210 DATA .1,0
220 DATA 4
230 DATA 1,-.1,0,1
240 DATA 1,0,20,1
250 DATA 1,-.0503,0,1
260 DATA 1,0,10,1
999 END
```

```
RUN
-.15
```

2

Exercise 23 – Description of Image

This exercise is a very good test of understanding of geometry and all the material that has preceded. The rules are given below:

Let the system matrix be described by four numbers as indicated.

$$M = \begin{bmatrix} a & b \\ c & d \end{bmatrix}$$

If a & c have the same signs, image is virtual.

If a & c have opposite signs, image is real.

If $d > 0$ & $b > 0$ then the image is erect.

If $d > 0$ & $b < 0$ & $|d| > |bc/a|$ then the image is erect.

If $d > 0$ & $b < 0$ & $|d| < |bc/a|$ then the image is inverted.

If $d < 0$ & $b < 0$ then the image is inverted.

If $d < 0$ & $b > 0$ & $|d| > |bc/a|$ then the image is inverted.

If $d < 0$ & $b > 0$ & $|d| < |bc/a|$ then the image is erect.

If $|d - bc/a| > 1$ then the image is magnified.

If $|d - bc/a| = 1$ then the image is the same size.

If $|d - bc/a| < 1$ then the image is reduced in size.